

Dissecting Different Flavors of Generic Programming in One Haskell Universe

Presented to Galois

Sean Leather

Utrecht University

August 27, 2013

What is Generic Programming?

In programming languages, the adjective “generic” is heavily overloaded.

- Java/C# generics
- C++ templates
- Ada generic packages

What is Generic Programming?

The goal is often the same.

A higher level of abstraction than “normally” available

The technique is also often similar.

Some form of parameterization and instantiation

Examples of Generic Programming

Java/C#:

```
public class Stack<T>
{
    public void push(T item) {...}
    public T pop() {...}
}
```

In other words:

- Java-style generics \approx parametric polymorphism

Examples of Generic Programming

C++:

```
template<typename T, typename Compare>
T& min(T& a, T& b, Compare comp) {
    if (comp(b, a))
        return b;
    return a;
}
```

In other words:

- C++ templates \approx ad-hoc polymorphism

Generic Programming in Haskell

“Generic programming”:

- For other languages, the term tends to be used for late additions.
- Parametric and ad-hoc polymorphism were available in Haskell from the beginning.

In Haskell, we have come to use “generic programming” for **datatype-generic programming** (a.k.a. “polytypism” or “shape/structure polymorphism”).

Datatype-Generic Programming

What is datatype-generic programming?

- Parameterize a function over the *structure* of datatypes
- Instantiate the function with a particular type

The result is a function that

- **works with many types** (polymorphism) but
- **uses knowledge of the type** (unlike parametric) and
- **need not be redefined for every type** (unlike ad-hoc).

Generic Functions

Applications

- Pretty-printing (e.g. `show`), parsing (e.g. `read`)
- Compression, serialization, marshallng (and their inverses)
- Comparison, equality
- (Co-)recursion, map, zip, zippers
- Traversals, queries, updates

Generic Platforms

Many different implementations:

- Preprocessors:
 - ▶ PolyP
 - ▶ Generic Haskell
- Libraries
 - ▶ Scrap Your Boilerplate (SYB) – included with GHC for a long time
 - ▶ Extensible and Modular Generics for the Masses (EMGM)
 - ▶ Regular – recursion schemes
 - ▶ Multirec – mutually recursive datatypes
 - ▶ Generic Deriving – available in GHC ≥ 7.2 , similar to Instant Generics
 - ▶ (and many, many more)

Generic Flavors

The implementations can be grouped into flavors depending on how they view the structure of a datatype.

Some flavors (or views):

Spine A constructor is a sequence of types.

Example: SYB

Sums-of-products A datatype is a collection of alternative tuples of types.

Example: Generic Deriving

Fixed point A datatype is a sums-of-products with recursive points.

Example: Multirec

Dissecting a Datatype: Sums-of-Products

```
data Tsum = A1 | A2
```

A datatype can have:

- **Alternatives:** unique constructors (≥ 0)

Dissecting a Datatype: Sums-of-Products

```
data Tprod = P2 Char Int
```

A datatype can have:

- **Fields**: types for each constructor (≥ 0)

Dissecting a Datatype: Sums-of-Products

Other features that are modeled:

- Constant types: each type in a field
- Parameters: type variables (≥ 0)

Features that are not modeled:

- Recursion
- Nesting (though it can be)

Modeling a Sum

To model (nested) alternatives:

```
data Either a b = Left a | Right b
```

For syntactic elegance:

```
data a :+: b = L a | R b
```

Modeling a Product

To model (nested) fields:

```
data (,) a b = (,) a b
```

For syntactic elegance:

```
data a :×: b = a :×: b
```

Modeling Other Structures

A constructor without fields:

```
data U = U
```

A constructor name:

```
data C a = C String a
```

A field type:

```
data K a = K a
```

Note: There are other features of datatypes, but we consider only the above.

Modeling an Example

An example datatype:

```
data E a = E1 | E2 a (E a) Int
```

The corresponding **structure representation type**:

```
type RepE a = C U :+: C (K a :×: K (E a) :×: K Int)
```

Notes:

- `:+:` is **infixr 5** and `:×:` is **infixr 6**.
- Operators nest to the right.

Converting Between Types: Isomorphism

- Generic functions work on the sums-of-products model.
- But first we need to convert between the model and the actual value of the datatype.
- We define an **isomorphism**: two total, dual functions.

```
fromE :: E a → RepE a
fromE E1           = L (C "E1" U)
fromE (E2 x e i) = R (C "E2" ((K x) :×: (K e) :×: (K i)))
```

```
toE :: RepE a → E a
toE (L (C "E1" U))           = E1
toE (R (C "E2" ((K x) :×: (K e) :×: (K i)))) = E2 x e i
```

Converting Between Types: Isomorphism

For convenience, we join the representation type and isomorphism in a type class `Generic` with an associated type synonym `Rep`.

```
class Generic a where  
  type Rep a  
  from :: a → Rep a  
  to   :: Rep a → a
```

The instance for `E`:

```
instance Generic (E a) where  
  type Rep (E a) = RepE a  
  from = fromE  
  to   = toE
```

Generic Functions

A **generic function**

- Is defined on each case of the structure representation and
- Works for every datatype that has a structure representation and isomorphism.

Example: `showRep a :: a → String`

- We will define a `show` function for each case.

Defining a Generic Function: `show`

Unit:

```
showU :: U → String  
showU U = ""
```

Constructor name:

```
showC :: (a → String) → C a → String  
showC showa (C nm a) = "(" ++ nm ++ " " ++ showa a ++ ")"
```

Field:

```
showK :: (a → String) → K a → String  
showK showa (K a) = showa a
```

Defining a Generic Function: `show`

Binary sum:

```
show+ :: (a → String) → (b → String) → a :+: b → String
show+ showa (L a) = showa a
show+ showb (R b) = showb b
```

Binary product:

```
show× :: (a → String) → (b → String) → a :×: b → String
show× showa showb (a :×: b) = showa a ++ " " ++ showb b
```

Defining a Generic Function: `show`

Recall:

```
type RepE a = C U :+: C (K a :×: K (E a) :×: K Int)
```

We can define a `show` function (assuming `showInt`):

```
showRepE :: (a → String) → ((a → String) → E a → String)  
           → RepE a → String
```

```
showRepE showa showE =  
  show+ (showC showU)  
        (showC (show× (showK showa)  
                      (show× (showK (showE showa)) (showK showInt))))))
```

Defining a Generic Function: `show`

```
showRepE :: (a → String) → ((a → String) → E a → String)
              → RepE a → String

showRepE showa showE =
  show+ (showC showU)
        (showC (show× (showK showa)
                      (show× (showK (showE showa)) (showK showInt))))))
```

The `showE` function itself is just an isomorphism away:

```
showE :: (a → String) → E a → String
showE showa = showRepE showa showE ∘ fromE
```


Defining a Generic Function: `show`

```
showRepE :: (a → String) → ((a → String) → E a → String)
           → RepE a → String

showRepE showa showE =
  show+ (showC showU)
        (showC (show× (showK showa)
                      (show× (showK (showE showa)) (showK showInt))))))
```

Some observations:

- This is **not** a generic function.
- It is defined on the structure of `E`, not on datatypes in general.
- It demonstrates a predictable pattern for defining the generic function.

Defining a Generic Function: `show`

```
showU ::      U      → String
showC :: ... ⇒ C a   → String
show+ :: ... ⇒ a :+: b → String
...
```

- The `show_...` functions can be thought of as recursive but not in the usual way because the argument types differ.
- **Polymorphic recursion** – functions with different types that have a common scheme that reference each other

Polymorphic Recursion

There are several ways to encode polymorphic recursion. We will use type classes.

- Standard classes already use polymorphic recursion for deriving instances: `Show`, `Eq`, etc.
- The class declaration specifies the type signature.
- Each recursive case is specified by an instance of the class.

A simplified definition of the `Show` class:

```
class Show a where  
  show :: a → String
```

Polymorphic Recursion

The instances for each structure representation case:

Unit:

```
instance Show U where  
  show = showU
```

Constructor name:

```
instance Show a  $\Rightarrow$  Show (C a) where  
  show = showC show
```

Binary product:

```
instance (Show a, Show b)  $\Rightarrow$  Show (a  $\times$  b) where  
  show = show $\times$  show show
```

Binary sum:

```
instance (Show a, Show b)  $\Rightarrow$  Show (a  $+$  b) where  
  show = show+ show show
```

Polymorphic Recursion

Now, recall `show_Rep_D` :

```
showRepE :: (a → String) → ((a → String) → E a → String)
           → RepE a → String

showRepE showa showE =
  show+ (showC showU)
        (showC (show× (showK showa)
                      (show× (showK (showE showa)) (showK showInt))))))
```

Compare to the new version that is now possible:

```
show_Rep_D' :: Show p ⇒ Rep_D' p → String
show_Rep_D' = show
```

Encoding Isomorphisms

To define the `show` function for `D`, we still need to define another function:

```
show_D' :: Show p => D p → String  
show_D' = show_Rep_D' ∘ from_D'
```

Next goal:

- Define one `show` function that knows how to convert any type `T` to its structure representation type `Rep_T`, given an isomorphism between `T` and `Rep_T`.

Encoding Isomorphisms

We define a class of function pairs.

- We again use a type class, but with the addition of a *type family*.
- Each function pair implements an isomorphism between a datatype T and its structure representation Rep_T :

$\text{from} :: T \rightarrow \text{Rep}_T$

$\text{to} :: \text{Rep}_T \rightarrow T$

- Each requires two types, so each instance must have two types (unlike the `Show` instances which needed only the structure representation type).
- Rep_T is precisely determined by T , so really we only need one unique type and a second type derivable from the first.
- In this case, a (1) multiparameter type class with a functional dependency and a (2) type class with a type family are equally expressive. (It's a matter of taste, really.)

Encoding Isomorphisms

The type class:

```
class Generic a where  
  type Rep a  
  from :: a → Rep a  
  to   :: Rep a → a
```

- `Rep` is a type family or, more precisely, an associated type synonym.
- Think of `Rep` as a function on types. Given a unique type (index) `T`, you get a type (synonym) `Rep T`.
- Note that `Rep T` need not be different from `Rep U` even though `T` and `U` are different.
- Concretely: two datatypes may have the same representation.

Encoding Isomorphisms

We need `Generic` instances for every datatype that we want to use with generic functions.

The instance for `D` uses definitions that we've already seen:

```
instance Generic (D p) where  
  type Rep (D p) = Rep_D' p  
  from = from_D'  
  to   = to_D'
```

- Other instances are defined similarly.
- In fact, `Rep T`, `from`, and `to` are precisely determined by the definition of `T`, so these instances can be automatically generated (e.g. using Template Haskell or a preprocessor).

The Generic `show` Function

Finally:

```
gshow :: (Show (Rep a), Generic a) => a -> String
gshow = show o from
```

GP in General

- Datatype-generic programming:
 - ▶ Datatype is the parameter
 - ▶ Instantiation gives you a large class of generic functions
- Many generic functions:
 - ▶ Pretty-printing (`show`) and parsing (`read`)
 - ▶ Compression, serialization, and the reverse
 - ▶ Comparison, equality
 - ▶ Folds (catamorphisms), unfolds (anamorphisms), maps, zips, zippers
 - ▶ Traversals, updates, queries
- Many different libraries:
 - ▶ Instant Generics – presented here
 - ▶ Generic Deriving – GHC \geq 7.2, similar to Instant Generics
 - ▶ EMGM – maintained by me
 - ▶ Regular – folds, etc.
 - ▶ Multirec – mutually recursive datatypes, folds, etc.
 - ▶ Scrap Your Boilerplate (SYB) – GHC, traversals, queries
 - ▶ ...

References

Generic Programming in Haskell:

- Johan Jeuring, Sean Leather, José Pedro Magalhães, Alexey Rodriguez Yakushev. *Libraries for Generic Programming in Haskell*. AFP 2008. pp. 165-229, 2009.
- Generic Deriving:
<http://www.haskell.org/haskellwiki/GHC.Generics>